TEMPLATE = app
TARGET = edgy
DEPENDPATH += yaml .
INCLUDEPATH += yaml .

# this is necessary for libyaml
DEFINES += HAVE_CONFIG_H
# uncomment the following if you want the old-style config
# DEFINES += OLD_CONFIG
OBJECTS_DIR = obj

LIBS += -lXtst -lstdc++ -lXcursor
CONFIG -= qt
CONFIG += x11
# Comment the following out if debug is needed
CONFIG += release

# Input
SOURCES += edgy.cc gfblib.cc 
!contains(DEFINES, OLD_CONFIG) {
	SOURCES += configdata.cc configparser.cc yaml/*.c
}
# following for testing purposes
#SOURCES += test.cc gfblib.cc configdata.cc configparser.cc yaml/*.c

# INSTALLPREFIX *= /usr/local
# the above is buggy, so use the isEmpty() function below
isEmpty(INSTALLPREFIX) {
	INSTALLPREFIX = /usr/local
}
BINPREFIX *= $${INSTALLPREFIX}/bin
DOCPREFIX *= $${INSTALLPREFIX}/share/doc/edgy

# Installs
binary.path = $${BINPREFIX}
binary.files = $${TARGET}

documentation.path = $${DOCPREFIX}
documentation.files = docs/*

# this has to come after the install defs
INSTALLS += binary documentation
